# F08AEFP (PDGEQRF)

## NAG Parallel Library Routine Document

**Note:** before using this routine, please read the Users' Note for your implementation to check for implementation-dependent details. You are advised to enclose any calls to NAG Parallel Library routines between calls to Z01AAFP and Z01ABFP.

# 1 Description

F08AEFP (PDGEQRF) computes the $QR$ factorization of a real $m$ by $n$ matrix $A_s$ (i.e., $A_s = QR$), where $A_s$ is a submatrix of a larger $m_A$ by $n_A$ matrix $A$, i.e.,

$$A_s(1:m, 1:n) \equiv A(i_A : i_A + m - 1, j_A : j_A + n - 1).$$

**Note:** if $i_A = j_A = 1$, $m = m_A$ and $n = n_A$, then $A_s = A$.

The orthogonal matrix $Q$ is not formed explicitly but is represented as a product of elementary reflectors

$$Q = H_1 H_2 \dots H_k, \quad \text{where} \quad k = \min(m, n).$$

Each elementary reflector $H_\ell$ has the form

$$H_\ell = I - \tau_\ell v_\ell v_\ell^T,$$

where $\tau_\ell$ is a real scalar, and $v_\ell$ is a real vector of length $m$. No pivoting is performed by F08AEFP (PDGEQRF).

# 2 Specification

```
SUBROUTINE F08AEFP(M, N, A, IA, JA, IDESCA, TAU, WORK, LWORK, INFO)
ENTRY      PDGEQRF(M, N, A, IA, JA, IDESCA, TAU, WORK, LWORK, INFO)
DOUBLE PRECISION   A(*), TAU(*), WORK(*)
INTEGER            M, N, IA, JA, IDESCA(*), LWORK, INFO
```

The ENTRY statement enables the routine to be called by its ScaLAPACK name.

# 3 Usage

## 3.1 Definitions

The following definitions are used in describing the data distribution within this document:

| | | |
|---|---|---|
| $m_p$ | – | the number of rows in the Library Grid. |
| $n_p$ | – | the number of columns in the Library Grid. |
| $p_r$ | – | the row grid coordinate of the calling processor. |
| $p_c$ | – | the column grid coordinate of the calling processor. |
| $M_b^X$ | – | the blocking factor for the distribution of the rows of a matrix $X$. |
| $N_b^X$ | – | the blocking factor for the distribution of the columns of a matrix $X$. |
| numroc($\alpha$,$b_\ell$,$q$,$s$,$k$) | – | a function which gives the **num**ber of **r**ows **o**r **c**olumns of a distributed matrix owned by the processor with the row or column coordinate $q$ ($p_r$ or $p_c$), where $\alpha$ is the total number of rows or columns of the matrix, $b_\ell$ is the blocking factor used ($M_b^X$ or $N_b^X$), $s$ is the row or column coordinate of the processor that possesses the first row or column of the distributed matrix and $k$ is either $m_p$ or $n_p$. The Library provides the function Z01CAFP (NUMROC) for the evaluation of this function. |
| indxg2p($i_g$,$b_\ell$,$q$,$s$,$k$) | – | a function which gives the processor row or column coordinate which possess the row or column index $i_g$ of the distributed full matrix $A$. The arguments $b_\ell, q, s$ and $k$ have the same meaning as in the function numroc. The Library provides the function Z01CDFP (INDXG2P) for the evaluation of this function. |

### 3.2 Global and Local Arguments

The following global **input** arguments must have the same value on entry to the routine on each processor and the global **output** arguments will have the same value on exit from the routine on each processor:

Global input arguments: M, N, IA, JA, IDESCA(1), IDESCA(3:8)

Global output arguments: INFO

The remaining arguments are local.

### 3.3 Distribution Strategy

The matrix $A$ must be partitioned into $M_b^A$ by $N_b^A$ rectangular blocks which are stored in an array A in a cyclic two-dimensional block distribution. This data distribution is described in more detail in the F08 Chapter Introduction. The array TAU is distributed across the processor columns in a cyclic two-dimensional block fashion, and is aligned with the rows of the matrix $A$.

### 3.4 Related Routines

This routine assumes that the data has already been correctly distributed, and if this is not the case will fail to produce correct results. The Library provides many support routines for the generation, scattering/gathering and input/output of matrices/vectors in cyclic two-dimensional block form. The following routines may be used in conjunction with F08AEFP (PDGEQRF):

Real matrix generation: F01ZQFP

Real matrix input: X04BCFP

Real matrix output: X04BDFP

Real matrix gather: F01WAFP

Real matrix scatter: F01WNFP

## 4 Arguments

**1:** M — INTEGER *Global Input*

*On entry:* $m$, the number of rows of the submatrix $A_s$.

*Constraint:* $0 \leq \mathrm{M} \leq \mathrm{IDESCA}(3)$.

**2:** N — INTEGER *Global Input*

*On entry:* $n$, the number of columns of the submatrix $A_s$.

*Constraint:* $0 \leq \mathrm{N} \leq \mathrm{IDESCA}(4)$.

**3:** A(∗) — DOUBLE PRECISION array *Local Input/Local Output*

**Note:** array A is formally defined as a vector. However, you may find it more convenient to consider A as a two-dimensional array of dimension (IDESCA(9),$\gamma$), where
$\gamma \geq$ numroc(JA+N−1, IDESCA(6), $p_c$, IDESCA(8), $n_p$).

*On entry:* the local part of the $m$ by $n$ matrix $A_s$ to be factorized.

*On exit:* if $m \geq n$, the elements below the diagonal of $A_s$ are overwritten by details of the orthogonal matrix $Q$ and the upper triangle is overwritten by the corresponding elements of the $n$ by $n$ upper triangular matrix $R$.
If $m < n$, the strictly lower triangular part of $A_s$ is overwritten by details of the orthogonal matrix $Q$ and the remaining elements are overwritten by the corresponding elements of the $m$ by $n$ upper trapezoidal matrix $R$.

**4:** IA — INTEGER *Global Input*

*On entry:* $i_A$, the row index of matrix $A$ that identifies the first row of the matrix $A_s$ to be factorized.

*Constraint:* $1 \leq \mathrm{IA} \leq \mathrm{IDESCA}(3) - \mathrm{M} + 1$.

**5:** JA — INTEGER *Global Input*

*On entry:* $j_A$, the column index of matrix $A$ that identifies the first column of the matrix $A_s$ to be factorized.

*Constraint:* $1 \leq \text{JA} \leq \text{IDESCA}(4) - \text{N} + 1$.

**6:** IDESCA(∗) — INTEGER array *Local Input*

**Note:** the dimension of the array IDESCA must be at least 9.

*Distribution:* the array elements IDESCA(1) and IDESCA(3),. . .,IDESCA(8) must be global to the processor grid and the elements IDESCA(2) and IDESCA(9) are local to each processor.

*On entry:* the description array for the matrix $A$. This array must contain details of the distribution of the matrix $A$ and the logical processor grid.

IDESCA(1), the descriptor type. For this routine, which uses a cyclic two-dimensional block distribution, IDESCA(1) = 1;
IDESCA(2), the Library context, usually returned by a call to the Library Grid initialisation routine Z01AAFP;
IDESCA(3), the number of rows, $m_A$, of the matrix $A$;
IDESCA(4), the number of columns, $n_A$, of the matrix $A$;
IDESCA(5), the blocking factor, $M_b^A$, used to distribute the rows of the matrix $A$;
IDESCA(6), the blocking factor, $N_b^A$, used to distribute the columns of the matrix $A$;
IDESCA(7), the processor row index over which the first row of the matrix $A$ is distributed;
IDESCA(8), the processor column index over which the first column of the matrix $A$ is distributed;
IDESCA(9), the leading dimension of the conceptual two-dimensional array A.

*Constraints:*

IDESCA(1) = 1;
IDESCA(3) $\geq$ 0; IDESCA(4) $\geq$ 0;
IDESCA(5) $\geq$ 1; IDESCA(6) $\geq$ 1;
$0 \leq \text{IDESCA}(7) \leq m_p - 1$; $0 \leq \text{IDESCA}(8) \leq n_p - 1$;
IDESCA(9) $\geq$ max(1,numroc(IDESCA(3),IDESCA(5),$p_r$,IDESCA(7),$m_p$)).

**7:** TAU(∗) — DOUBLE PRECISION array *Local Output*

**Note:** the dimension of the array TAU must be at least $\alpha$, where
$\alpha$ = numroc($\beta$,IDESCA(6),$p_c$,IDESCA(8),$n_p$) and $\beta$ = JA + min(M,N) − 1.

*On exit:* the scalar factors $\tau_\ell$ of the elementary reflectors $H_\ell$.

**8:** WORK(∗) — DOUBLE PRECISION array *Local Workspace/Local Output*

**Note:** the dimension of WORK must be at least max(1,LWORK). The minimum value of LWORK required to successfully call this routine can be obtained by setting LWORK = −1. The required size is returned in array element WORK(1).

*On exit:* WORK(1) contains the minimum dimension of the array WORK required to successfully complete the task.

**9:** LWORK — INTEGER *Local Input*

*On entry:* either −1 (see WORK) or the dimension of the array WORK required to successfully complete the task. If LWORK is set to −1 on entry this routine simply performs some initial error checking and then, if these checks are successful, calculates the minimum size of LWORK required.

*Constraints:*

either LWORK = −1,
or LWORK $\geq$ IDESCA(6) $\times$ ($c_1 + c_2$+IDESCA(6)), where

$c_1$ = numroc(M+$d_1$,IDESCA(5),$p_r$,$e_1$,$m_p$);
$c_2$ = numroc(N+$d_2$,IDESCA(6),$p_c$,$e_2$,$n_p$);
$d_1$ = mod(IA−1,IDESCA(5));
$d_2$ = mod(JA−1,IDESCA(6));
$e_1$ = indxg2p(IA,IDESCA(5),$p_r$,IDESCA(7),$m_p$);
$e_2$ = indxg2p(JA,IDESCA(6),$p_c$,IDESCA(8),$n_p$).

**10:** INFO — INTEGER *Global Output*

The NAG Parallel Library provides a mechanism, via the routine Z02EAFP, to reduce the amount of parameter validation performed by this routine. For a full description refer to the Z02 Chapter Introduction.

*On exit:* INFO = 0 (or −9999 if reduced error checking is enabled) unless the routine detects an error (see Section 5).

# 5 Errors and Warnings

If INFO < 0 an explanatory message is output and control returned to the calling program.

INFO < 0

On entry, one of the arguments was invalid:

if the $k$th argument is a scalar INFO = $-k$;
if the $k$th argument is an array and its $j$th element is invalid, INFO = $-(100 \times k + j)$.

This error occurred either because a global argument did not have the same value on all logical processors, or because its value on one or more processors was incorrect. An explanatory message distinguishes between these two cases.

# 6 Further Comments

The total number of floating-point operations is approximately $\frac{2}{3}n^2(3m-n)$ if $m \geq n$ or $\frac{2}{3}m^2(3n-m)$ if $m < n$. To solve the system of equations $A_sX = B_s$, this routine may be followed by a call to F08AFFP (PDORGQR) which forms $Q$ explicitly. In terms of the $QR$ factorization, the linear system $A_sX = B_s$ is given by $QRX = B_s$, and hence $RX = Q^TB_s$. If $R$ is triangular then $X$ may be computed using the PBLAS routine PDTRSM. F08AGFP (PDORMQR) may be used to form $Q^TB_s$.

## 6.1 Algorithmic Detail

For a general $m$ by $n$ matrix $A$, if $m \geq n$, the factorization is given by:

$$A = Q \begin{pmatrix} R \\ 0 \end{pmatrix}$$

where $R$ is an $n$ by $n$ upper triangular matrix and $Q$ is an $m$ by $m$ orthogonal matrix. It is sometimes more convenient to write the factorization as

$$A = \begin{pmatrix} Q_1 & Q_2 \end{pmatrix} \begin{pmatrix} R \\ 0 \end{pmatrix}$$

which reduces to

$$A = Q_1R,$$

where $Q_1$ consists of the first $n$ columns of $Q$, and $Q_2$ the remaining $m-n$ columns.

If $m < n$, $R$ is upper trapezoidal, and the factorization can be written

$$A = Q \begin{pmatrix} R_1 & R_2 \end{pmatrix}$$

where $R_1$ is an $m$ by $m$ upper triangular matrix and $R_2$ is an $m$ by $n$ rectangular matrix.

The matrix $Q$ is not formed explicitly but is represented as a product of $\min(m,n)$ elementary reflectors. See Anderson *et al.* [1] and Blackford *et al.* [2] for details of the block algorithm used by the routine.

### 6.2 Parallelism Detail

The Level-3 BLAS operations are carried out in parallel within the routine.

### 6.3 Accuracy

The computed factorization is the exact factorization of a nearby matrix $A + E$, where

$$\|E\|_2 = \epsilon p(m,n)\|A\|_2,$$

$\epsilon$ is ***machine precision*** and $p(m,n)$ is a modest function of $m$ and $n$.

## 7 References

**[1]** Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia

**[2]** Blackford L S, Choi J, Cleary A, D'Azevedo E, Demmel J, Dhillon I, Dongarra J, Hammarling S, Henry G, Petitet A, Stanley K, Walker D and Whaley R C (1997) ScaLAPACK Users' Guide *SIAM* 3600 University City Science Center, Philadelpia, PA 19104-2688, USA. URL: http://www.netlib.org/scalapack/slug/scalapack_slug.html

**[3]** Golub G H and van Loan C F (1996) *Matrix Computations* Johns Hopkins University Press (3rd Edition), Baltimore

## 8 Example

To solve the linear least-squares problem

$$\min \|Ax^{(i)} - c^{(i)}\|_2 \quad \text{for} \quad i = 1, 2$$

where $c^{(1)}$ and $c^{(2)}$ are the columns of the matrix $C$,

$$A = \begin{pmatrix} -0.57 & -1.28 & -0.39 & 0.25 \\ -1.93 & 1.08 & -0.31 & -2.14 \\ 2.30 & 0.24 & 0.40 & -0.35 \\ -1.93 & 0.64 & -0.66 & 0.08 \\ 0.15 & 0.30 & 0.15 & -2.13 \\ -0.02 & 1.03 & -1.43 & 0.50 \end{pmatrix} \text{ and } \quad C = \begin{pmatrix} -3.15 & 2.19 \\ -0.11 & -3.64 \\ 1.99 & 0.57 \\ -2.70 & 8.23 \\ 0.26 & -6.35 \\ 4.50 & -1.48 \end{pmatrix}.$$

The example uses a 2 by 2 logical processor grid and a 2 by 2 block for both $A$ and $C$.

**Note:** the listing of the Example Program presented below does not give a full pathname for the data file being opened, but in general the user must give the full pathname in this and any other OPEN statement.

### 8.1 Example Text

```
*     F08AEFP Example Program Text
*     NAG Parallel Library Release 3 Revised. NAG Copyright 1999.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          DT
      PARAMETER        (DT=1)
      INTEGER          MB, NB
      PARAMETER        (MB=2,NB=MB)
      INTEGER          MMAX, NMAX, LDA, RHSMAX, LDC, IAROW, IACOL,
     +                 ICROW, ICCOL, LWORK, LIWORK
      PARAMETER        (MMAX=6,NMAX=4,LDA=MMAX,RHSMAX=2,LDC=MMAX,
```

```
     +                IAROW=0,IACOL=0,ICROW=0,ICCOL=0,LWORK=100,
     +                LIWORK=100)
*     .. Local Scalars ..
      DOUBLE PRECISION RCOND
      INTEGER          I, IA, IC, ICNTXT, IFAIL, INFO, J, JA, JC, M, MP,
     +                 N, NP, NRHS
      LOGICAL          ROOT
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), C(LDC,RHSMAX), CG(NMAX,RHSMAX),
     +                 TAU(MMAX), WORK(LWORK)
      INTEGER          IDESCA(9), IDESCC(9),
     +                 IWORK(LIWORK)
*     .. External Functions ..
      DOUBLE PRECISION X02AJF
      LOGICAL          Z01ACFP
      EXTERNAL         X02AJF, Z01ACFP
*     .. External Subroutines ..
      EXTERNAL         F01WAFP, F07TGFP, F08AEFP, F08AGFP, PDTRSM,
     +                 X04BCFP, Z01AAFP, Z01ABFP
*     .. Executable Statements ..
      ROOT = Z01ACFP()
      IF (ROOT) THEN
         WRITE (NOUT,*) 'F08AEFP Example Program Results'
         WRITE (NOUT,*)
      END IF
*
      MP = 2
      NP = 2
      IFAIL = 0
      CALL Z01AAFP(ICNTXT,MP,NP,IFAIL)
*
      OPEN (NIN,FILE='f08aefpe.d')
*
*     Skip heading in data file
*
      READ (NIN,*)
      READ (NIN,*) M, N, NRHS
*
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. NRHS.LE.RHSMAX) THEN
*
*        Set the starting address and array descriptor for A
*
         IA = 1
         JA = 1
         IDESCA(1) = DT
         IDESCA(2) = ICNTXT
         IDESCA(3) = M
         IDESCA(4) = N
         IDESCA(5) = MB
         IDESCA(6) = NB
         IDESCA(7) = IAROW
         IDESCA(8) = IACOL
         IDESCA(9) = LDA
*
*        Read the matrix A from data file
*
         IFAIL = 0
         CALL X04BCFP(NIN,M,N,A,IA,JA,IDESCA,IFAIL)
```

```
*
*        Set the starting address and array descriptor for C
*
         IC = 1
         JC = 1
         IDESCC(1) = DT
         IDESCC(2) = ICNTXT
         IDESCC(3) = M
         IDESCC(4) = NRHS
         IDESCC(5) = MB
         IDESCC(6) = NB
         IDESCC(7) = ICROW
         IDESCC(8) = ICCOL
         IDESCC(9) = LDC
*
*        Read the matrix C from data file
*
         IFAIL = 0
         CALL X04BCFP(NIN,M,NRHS,C,IC,JC,IDESCC,IFAIL)
*
*        Factorize A
*
         CALL F08AEFP(M,N,A,IA,JA,IDESCA,TAU,WORK,LWORK,INFO)
*
         IF (ROOT) THEN
            WRITE (NOUT,*) 'F08AEFP INFO = ', INFO
            WRITE (NOUT,*)
         END IF
         IF (INFO.NE.0) GO TO 40
*
*        Compute the solution
*
*        First apply Q to the right hand sides
*
         CALL F08AGFP('L','T',M,NRHS,N,A,IA,JA,IDESCA,TAU,C,IC,JC,
     +                IDESCC,WORK,LWORK,INFO)
*
         IF (ROOT) THEN
            WRITE (NOUT,*) 'F08AGFP INFO = ', INFO
            WRITE (NOUT,*)
         END IF
         IF (INFO.NE.0) GO TO 40
*
*        Compute (reciprocal of) the condition number of the upper
*        triangular matrix R
*
         CALL F07TGFP('1-norm','Upper','Non-unit',N,A,IA,JA,IDESCA,
     +                RCOND,WORK,LWORK,IWORK,LIWORK,INFO)
*
         IF (ROOT) THEN
            WRITE (NOUT,*) 'F07TGFP INFO = ', INFO
            WRITE (NOUT,*)
         END IF
         IF (INFO.NE.0) GO TO 40
*
*        Check that the condition number is no larger than 1/eps, where
*        eps is the machine precision
*
```

```
         IF (RCOND.GT.X02AJF()) THEN
*
*           Solve the triangular system
*
            CALL PDTRSM('Left','Upper','No transpose','Non-unit',N,NRHS,
     +                  1.0D0,A,IA,JA,IDESCA,C,IC,JC,IDESCC)
*
*           Gather the solution matrix C to the (root) processor (0,0)
*           where it is denoted by CG
*
            IFAIL = 0
            CALL F01WAFP(N,NRHS,C,IC,JC,IDESCC,0,0,CG,N,WORK,LWORK,
     +                   IFAIL)
*
*           Print the matrix CG
*
            IF (ROOT) THEN
               WRITE (NOUT,*) 'The least-squares solution'
               WRITE (NOUT,*)
               DO 20 I = 1, N
                  WRITE (NOUT,'(1X,4(F12.4,1X))') (CG(I,J),J=1,NRHS)
   20          CONTINUE
            END IF
         ELSE
            IF (ROOT) WRITE (NOUT,*)
     +          'Matrix is singular to working precision'
         END IF
*
      END IF
*
   40 CONTINUE
      CLOSE (NIN)
*
      IFAIL = 0
      CALL Z01ABFP(ICNTXT,'N',IFAIL)
*
      STOP
      END
```

## 8.2 Example Data

```
F08AEFP Example Program Data
 6 4 2                    :Values of M, N and NRHS
-0.57 -1.28 -0.39  0.25
-1.93  1.08 -0.31 -2.14
 2.30  0.24  0.40 -0.35
-1.93  0.64 -0.66  0.08
 0.15  0.30  0.15 -2.13
-0.02  1.03 -1.43  0.50  :End of matrix A
-3.15  2.19
-0.11 -3.64
 1.99  0.57
-2.70  8.23
 0.26 -6.35
 4.50 -1.48              :End of matrix C
```

## 8.3 Example Results

```
F08AEFP Example Program Results

F08AEFP INFO =            0

F08AGFP INFO =            0

F07TGFP INFO =            0

The least-squares solution

      1.5146      -1.5838
      1.8621       0.5536
     -1.4467       1.3491
      0.0396       2.9600
```